

Appendix I. A BASIC Computer Program for Calculating Interatomic Distances and Angles and Torsion Angles from Crystal Coordinates

The program transforms the crystal coordinates to Cartesian coordinates by applying Eqs. 5.30 and finds the torsion angle $\omega(\text{ABCD}) = \omega(\text{IJKL})$ by applying Eqs. 5.35 and 5.36. The input consists of the unit-cell dimensions (line 20) and the crystal coordinates of the atoms, which are assigned code numbers $I = 1$ to N (line 46). The input is terminated by writing $I = 0$. Each torsion angle required has to be requested separately by specifying the code numbers of the four atoms concerned: interatomic distances and angles are then printed in addition to the torsion angle. If torsion angles are not required, the code numbers of only three atoms need to be specified; if only a distance is required then code numbers of only two atoms suffice. A sample output follows.

App. I

Program

```

4 PRINT
5 PRINT "MOLECULAR GEOMETRY"
10 DIM X[60],Y[60],Z[60]
15 PRINT "A,B,C,ALPHA,BETA,GAMMA"
20 INPUT A1,A2,A3,w1,w2,w3
21 PRINT
25 LET P=3.14159/180
30 LET C1= COS (w1*P)
31 LET C2= COS (w2*P)
32 LET C3= COS (w3*P)
33 LET S3= SIN (w3*P)
35 LET M6=(C1-C3*C2)/S3
37 LET V= SQR (1-C1*C1-C2*C2-C3*C3+2*C1*C2*C3)
39 LET M9=V/S3
45 PRINT "I,X(I),Y(I),Z(I), I=0 FOR LAST ATOM"
46 INPUT I,X[I],Y[I],Z[I]
47 PRINT
48 IF I= 0 GOTO 75
49 LET U=A1*X[I]+A2*Y[I]*C3+A3*Z[I]*C2
50 LET V=A2*Y[I]*S3+A3*Z[I]*M6
51 LET W=A3*Z[I]*M9
52 LET X[I]=U
55 LET Y[I]=V
57 LET Z[I]=W
58 GOTO 46
75 PRINT "BOND LENGTH: I1,I2,0,0"
76 PRINT "BOND ANGLE : I1,I2,I3,0"
77 PRINT "TORSION ANGLE: I1,I2,I3,I4"
85 INPUT T1,T2,T3,T4
86 PRINT

```

Appendix I

```
100 LET E[1]=X[T2]-X[T1]
105 LET E[2]=Y[T2]-Y[T1]
110 LET E[3]=Z[T2]-Z[T1]
115 LET F=1
120 LET D1= SQR (E[1]*E[1]+E[2]*E[2]+E[3]*E[3])
121 LET D=D1
125 GOSUB 325
133 IF T3= 0 GOTO 499
135 LET E[4]=X[T3]-X[T2]
140 LET E[5]=Y[T3]-Y[T2]
145 LET E[6]=Z[T3]-Z[T2]
150 LET F=4
155 LET D2= SQR (E[4]*E[4]+E[5]*E[5]+E[6]*E[6])
156 LET D=D2
160 GOSUB 325
165 LET C4=-(E[1]*E[4]+E[2]*E[5]+E[3]*E[6])
170 LET S4= SQR (1-C4*C4)
175 LET A4= ATN (S4/C4)/P
176 IF A4> 0 GOTO 180
177 LET A4=A4+180
180 IF T4= 0 GOTO 497
185 LET E[7]=X[T4]-X[T3]
190 LET E[8]=Y[T4]-Y[T3]
195 LET E[9]=Z[T4]-Z[T3]
200 LET F=7
205 LET D3= SQR (E[7]*E[7]+E[8]*E[8]+E[9]*E[9])
210 LET D=D3
215 GOSUB 325
220 LET C5=-(E[4]*E[7]+E[5]*E[8]+E[6]*E[9])
225 LET S5= SQR (1-C5*C5)
230 LET A5= ATN (S5/C5)/P
231 IF A5> 0 GOTO 250
232 LET A5=A5+180
250 LET U1=(E[2]*E[6])-(E[3]*E[5])
252 LET U2=(E[3]*E[4])-(E[1]*E[6])
254 LET U3=(E[1]*E[5])-(E[2]*E[4])
256 LET V1=(E[5]*E[9])-(E[6]*E[8])
258 LET V2=(E[6]*E[7])-(E[4]*E[9])
260 LET V3=(E[4]*E[8])-(E[5]*E[7])
265 LET C6=(U1*V1+U2*V2+U3*V3)/(S4*S5)
270 LET S6=(E[1]*V1+E[2]*V2+E[3]*V3)/(S4*S5)
275 LET A6= ATN (S6/C6)/P
280 IF A6> 0 GOTO 290
281 IF S6< 0 GOTO 300
282 LET A6=A6+180
284 GOTO 300
290 IF C6> 0 GOTO 300
292 LET A6=A6-180
299 PRINT
300 PRINT " T('T1;T2;T3;T4;') ="A6
301 PRINT " W('T2;T3;T4;') ="A5
302 PRINT " D('T3;T4;') ="D3
305 GOTO 497
325 FOR J=F TO F+2
327 LET E[J]=E[J]/D
329 NEXT J
331 RETURN
496 PRINT
497 PRINT " W('T1;T2;T3;') ="A4
498 PRINT " D('T2;T3;') ="D2
499 PRINT " D('T1;T2;') ="D1
510 PRINT
```

```
512 GOTO 85
515 END
```

App. I

Sample input and output

BASIC?

*READY

RUN

MOLECULAR GEOMETRY

A,B,C,ALPHA,BETA,GAMMA

? 7.877? 7.210? 7.891? 105.56? 116.25? 79.84

I,X(I),Y(I),Z(I), I=0 FOR LAST ATOM

? 1? -.1361? .1572? -.0717

? 2? -.0963? .1157? .1184

? 3? .0920? .0109? .2083

? 4? -.1955? .3344? -.1057

? 5? -.2272? .4046? -.2806

? 6? -.2182? .1553? .2003

? 7? -.4167? .2445? .1235

? 8? .2174? .0656? .3903

? 9? .2034? .2392? .5396

? 10? .1361? -.1572? .0717

? 11? .0963? -.1157? -.1184

? 12? -.0920? -.0109? -.2083

? 0? 0? 0? 0

BOND LENGTH: I1,I2,0,0

BOND ANGLE : I1,I2,I3,0

TORSION ANGLE: I1,I2,I3,I4

? 12? 1? 2? 3

T(12 1 2 3) = 46.2811

W(1 2 3) = 114.158

D(2 3) = 1.49519

W(12 1 2) = 114.083

D(1 2) = 1.49324

D(12 1) = 1.49442

? 1? 2? 3? 10

T(1 2 3 10) = -46.3004

W(2 3 10) = 114.124

D(3 10) = 1.49442

W(1 2 3) = 114.158

D(2 3) = 1.49519

D(1 2) = 1.49324

? 2? 3? 10? 11

T(2 3 10 11) = 46.2653

W(3 10 11) = 114.083

D(10 11) = 1.49324

W(2 3 10) = 114.124

D(3 10) = 1.49442

D(2 3) = 1.49519