Appendix II. A BASIC Computer Program for Calculating Cartesian Coordinates from Internal Coordinates

Bond distances, bond angles, and torsion angles are the internal coordinates used. Care is required to ensure that the internal coordinates chosen are geometrically independent and that they fix the structure of the molecular skeleton completely. This is done by regarding the molecule (N atoms) as a main chain with an arbitrary number of branches. Its structure is then completely fixed by specifying $3N-6$ internal coordinates consisting of $N-1$ bond distances, $N-2$ bond angles, and $N-3$ torsion angles.

The program assigns sequential code numbers $I = 1$ to 100 to the atoms in the order they are entered in the input list. For each atom entered, the input consists of the code number J of the atom to which it is directly linked (the "next atom" but always such that $J < I$), together with the values of the bond angle $W(I,J,J-1)$ for $I \geq 3$, the signed torsion angle $T(I,J,J-1,J-2)$ for $I \geq 4$, and the bond distance $R(I,J)$ for $I \geq 2$, in that order. For $I = 1$, $J = 0$; the six undefined quantities involving this nonexistent atom—e.g., $R(1,0)$, $W(2,1,0)$, $T(3,2,1,0)$, etc., are set equal to zero (see sample input and output). Angles are in degrees.

The origin and axial orientation of the Cartesian frame are tied to the first three atoms entered ($I = 1,2,3$). The origin is at atom 1, the X-axis is along the direction from atom 2 to atom 1, and the Y-axis is in the plane of atoms 1, 2, and 3. Thus, the Cartesian coordinates of these three atoms are:

	x	y	z
1	0	0	0
2	$-r_{2,1}$	0	0
3	$-r_{2,1}+r_{3,2}\cos\omega_{3,2,1}$	$r_{3,2}\sin\omega_{3,2,1}$	0

For each subsequent atom entered, a local Cartesian frame (X_I,Y_I,Z_I) is set up in a similar way, with its origin at the new atom I, X_I along the direction from J to I, and Y_I in the plane of $I,J,J-1$. The coordinates of a point in the Ith system are related to the coordinates in the Jth system by a rotational transformation plus a translation[1]

[1] H. B. Thompson, *J. Chem. Phys.* **47**, 3407 (1967).

$$\begin{bmatrix} x_J \\ y_J \\ z_J \\ 1 \end{bmatrix} = \begin{bmatrix} -\cos W & -\sin W & 0 & -R\cos W \\ \sin W\cos T & -\cos W\cos T & -\sin T & R\sin W\cos T \\ \sin W\sin T & -\cos W\sin T & \cos T & R\sin W\sin T \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} x_I \\ y_I \\ z_I \\ 1 \end{bmatrix}$$

$$\text{or } \mathbf{x}_J = \mathbf{B}_I \cdot \mathbf{x}_I$$

writing $\mathbf{x}_I$ for the expanded vector and $\mathbf{B}_I$ for the matrix. Transformation back to the initial frame is then performed by matrix multiplication:

$$\mathbf{x}_1 = \mathbf{B}_2 \cdot \mathbf{B}_3 \cdot \mathbf{B}_4 \ldots . \mathbf{B}_I \cdot \mathbf{x}_I$$

To ensure self-consistency, some of the elements of the first two matrices $\mathbf{B}_2$ and $\mathbf{B}_3$ are fixed by the program, but all other matrix elements depend on the input.

App. II Program

```
9  DIM A[8,100],Q[2,100]
10  PRINT
11  PRINT
15  PRINT "TRANSFORMATION OF INTERNAL TO CARTESIAN COORDINATES"
20  PRINT "INPUT SEQUENCE IS: NEXT,BOND ANGLE, TORSION,DISTANCE"
21  PRINT "INPUT NEXT=999 TO STOP"
22  PRINT "NUMBERING OF ATOMS HAS TO START WITH ZERO"
23  PRINT
24  PRINT
25  LET I= 0
26  LET F=1.74533E-2
30  FOR S=1 TO 7
35    LET A[S, 0]= 0
40  NEXT S
45  LET A[ 0, 0]=1
50  LET A[4, 0]=1
55  LET A[8, 0]=1
60  LET Q[ 0, 0]= 0
65  LET Q[1, 0]= 0
70  LET Q[2, 0]= 0
75  PRINT "NEXT ATOM";
76  INPUT N
77  IF N=999 GOTO  299
78  INPUT W,T,R
79  IF I= 0 GOTO  261
95  IF I>2 GOTO  140
100  IF I<2 GOTO  120
105  LET T= 0
110  LET M=1
115  GOTO  155
120  LET W= 0
125  LET T=180-T
130  LET M= 0
135  GOTO  155
140  LET M=N-1
155  LET N=M
160  LET B0=- COS (W*F)
165  LET B1=- SIN (W*F)
170  LET B5=- SIN (T*F)
175  LET B8= COS (T*F)
```

Appendix II

```
180  LET B3=-B1*B8
185  LET B4=B0*B8
190  LET B6=B1*B5
195  LET B7=-B0*B5
200  LET B2= 0
205  LET A( 0,I)=A( 0,M)*B0+A(1,M)*B3+A(2,M)*B6
210  LET A(1,I)=A( 0,M)*B1+A(1,M)*B4+A(2,M)*B7
215  LET A(2,I)=A( 0,M)*B2+A(1,M)*B5+A(2,M)*B8
220  LET A(3,I)=A(3,M)*B0+A(4,M)*B3+A(5,M)*B6
225  LET A(4,I)=A(3,M)*B1+A(4,M)*B4+A(5,M)*B7
230  LET A(5,I)=A(3,M)*B2+A(4,M)*B5+A(5,M)*B8
235  LET A(6,I)=A(6,M)*B0+A(7,M)*B3+A(8,M)*B6
240  LET A(7,I)=A(6,M)*B1+A(7,M)*B4+A(8,M)*B7
245  LET A(8,I)=A(6,M)*B2+A(7,M)*B5+A(8,M)*B8
250  LET Q( 0,I)=Q( 0,M)+A( 0,I)*R
255  LET Q(1,I)=Q(1,M)+A(3,I)*R
260  LET Q(2,I)=Q(2,M)+A(6,I)*R
261  PRINT
262  LET I1=I+1
265  PRINT I1,Q( 0,I),Q(1,I),Q(2,I)
266  PRINT
270  LET I=I+1
275  GOTO  75
299  PRINT
300  PRINT "TYPE 1 IF NEW INPUT SET"
305  PRINT "TYPE 0 IF END";
310  INPUT Z
315  IF Z=1 GOTO  10
320  END
```

App. II Sample input and output

```
TRANSFORMATION OF INTERNAL TO CARTESIAN COORDINATES
INPUT SEQUENCE IS: NEXT,BOND ANGLE, TORSION,DISTANCE
INPUT NEXT=999 TO STOP
NUMBERING OF ATOMS HAS TO START WITH ZERO


NEXT ATOM? 0? 0? 0? 0
 1              0               0               0

NEXT ATOM? 1? 0? 0? 1.525
 2             -1.525           0               0

NEXT ATOM? 2? 107.12? 0? 1.531
 3             -1.97568         1.46316         0

NEXT ATOM? 3? 104.08? 28.5? 1.518
 4             -.847761         2.19699        -.702565

NEXT ATOM? 4? 100.50? -33.7? 1.542
 5              .362545         1.36111        -.239745

NEXT ATOM? 4? 109.71? 91.6? 1.535
 6             -1.14158         2.27922        -2.20693

NEXT ATOM? 4? 112.82? -148.5? 1.529
 7             -.659971         3.62536        -.190399
```

```
NEXT ATOM? 999
TYPE 1 IF NEW INPUT SET
TYPE Ø IF END? Ø

*READY
```

The sample calculation refers to the skeleton



with the following input parameters:

I	J	W	T	R
2	1	0	0	1.525
3	2	107.12	0	1.531
4	3	104.08	+28.5	1.518
5	4	100.50	−33.7	1.542
6	4	109.71	+91.6	1.535
7	4	112.82	−148.5	1.529

From the output, the Cartesian coordinates are:

I	x	y	z
1	0	0	0
2	−1.525	0	0
3	−1.976	1.463	0
4	−0.848	2.197	−0.703
5	0.363	1.361	−0.240
6	−1.142	2.279	−2.207
7	−0.660	3.625	−0.190

from which the dependent bond lengths and angles may readily be calculated (e.g., using the program of Appendix I).